
Subject: Use slider from layout designer?
Posted by lectus on Tue, 17 Oct 2006 19:56:12 GMT
View Forum Message <> Reply to Message

Will Slider be available from the layout designer?

Subject: Re: Use slider from layout designer?
Posted by mirek on Sat, 21 Oct 2006 01:27:38 GMT
View Forum Message <> Reply to Message

lectus wrote on Tue, 17 October 2006 15:56Will Slider be available from the layout designer?

You can place Slider or any class that does not have .usc description by using "User class" and
typing the class name manually.

Sure, .usc description would be useful as well, but right now is not top priority. Maybe the
community could help with .usc scripts (it really is not that complicated).

Mirek

Subject: Re: Use slider from layout designer?
Posted by victorb on Tue, 13 Mar 2007 22:02:22 GMT
View Forum Message <> Reply to Message

Here is an update of CtrlLib.usc for SliderCtrl

ctrl SliderCtrl {
	group "Progress";

	GetMinSize() { sz = GetTextSize("X"); sz.cy += 2; return sz; }
	GetStdSize() { sz = GetMinSize(); sz.cx *= 7; return sz; }

	Frame  SetFrame @1;

	Paint(w) {
		r = GetRect();
		
		sz = Size(r.right - r.left, r.bottom - r.top);
		
		if (sz.cx < sz.cy) {
			halfX = sz.cx >> 1;	
			halfY = sz.cy >> 1;
			DrawInsetFrame(w, Rect(halfX - 2, 2, halfX + 2, sz.cy - 2));
			imgSz = GetImageSize("CtrlImg::vthumb");			
			w.DrawImage(halfX - (imgSz.cx >> 1), halfY - (imgSz.cy >> 1), "CtrlImg::vthumb");

		} else {
			halfX = sz.cx >> 1;	
			halfY = sz.cy >> 1;
			DrawInsetFrame(w, Rect(2, halfY - 2, sz.cx - 2, halfY + 2));
			imgSz = GetImageSize("CtrlImg::hthumb");			
			w.DrawImage(halfX - (imgSz.cx >> 1), halfY - (imgSz.cy >> 1), "CtrlImg::hthumb");			
		}					
	}
}

Could someone please add it to the main code tree ?

Thanks,
Victor

Subject: Re: Use slider from layout designer?
Posted by victorb on Tue, 13 Mar 2007 23:17:29 GMT
View Forum Message <> Reply to Message

Please discard my previous message... The following code now works with Ctrl frames. I have
also included a "basic" ProgressIndicator. They should be ready for integration in the main
branch.

ctrl SliderCtrl {
	group "Progress";

	GetMinSize() { return Size(0, 0); }
	GetStdSize() { return Size(64, 24); }

	Frame  SetFrame @1;

	Paint(w) {
		r = GetRect();
		DrawCtrlFrame(w, r, .SetFrame);
		
		sz = Size(r.right - r.left, r.bottom - r.top);

		halfX = int(r.left + r.right) >> 1;
		halfY = int(r.top + r.bottom) >> 1;
		
		if (sz.cx < sz.cy) {
			DrawInsetFrame(w, Rect(halfX - 2, r.top + 2, halfX + 2, r.bottom - 2));
			imgSz = GetImageSize("CtrlImg::vthumb");			
			w.DrawImage(halfX - (imgSz.cx >> 1), halfY - (imgSz.cy >> 1), "CtrlImg::vthumb");
		} else {
			DrawInsetFrame(w, Rect(r.left + 2, halfY - 2, r.right - 2, halfY + 2));
			imgSz = GetImageSize("CtrlImg::hthumb");			

			w.DrawImage(halfX - (imgSz.cx >> 1), halfY - (imgSz.cy >> 1), "CtrlImg::hthumb");			
		}					
	}
}

ctrl ProgressIndicator {
	group "Progress";

	GetMinSize() { return Size(0, 0); }
	GetStdSize() { return Size(64, 24); }

	Frame  SetFrame @1;

	Paint(w) {
		r = GetRect();

		DrawCtrlFrame(w, r, .SetFrame);	
		DrawInsetFrame(w, r);
		
		sz = Size(r.right - r.left, r.bottom - r.top);
		
		w.DrawRect(r.left, r.top, sz.cx, sz.cy, :SBlack);
		DeflateRect(r);
		sz = Size(r.right - r.left, r.bottom - r.top);
		w.DrawRect(r.left, r.top, sz.cx, sz.cy, :SWhite);		

		if(sz.cx > sz.cy) {
			w.DrawRect(r.left, r.top, sz.cx >> 2, sz.cy, :SLtGreen);		
		}
		else {
			w.DrawRect(r.left, r.bottom - (sz.cy >> 2), sz.cx, sz.cy >> 2, :SLtGreen);
		}
	}
}

Writting usc is very easy... when you understand how to deal with frames   

Subject: Re: Use slider from layout designer?
Posted by mirek on Wed, 14 Mar 2007 14:00:39 GMT
View Forum Message <> Reply to Message

Thank you, now it the main tree!'

Mirek

